<commit_before>{ lib
, stdenv
, fetchFromGitHub
, fetchzip
, mkYarnPackage
, baseUrl ? null
}:

mkYarnPackage rec {
  pname = "synapse-admin";
  version = "0.8.5";
  src = fetchFromGitHub {
    owner = "Awesome-Technologies";
    repo = pname;
    rev = version;
    sha256 = "sha256-0miHtEJ5e8MaqGc4ezPvwhGjoCZyOE7md0DUCC/ZOfk=";
  };

  yarnNix = ./yarn.nix;
  yarnLock = ./yarn.lock;
  packageJSON = ./package.json;

  NODE_ENV = "production";
  ${if baseUrl != null then "REACT_APP_SERVER" else null} = baseUrl;

  buildPhase = ''
    runHook preBuild

    export HOME=$(mktemp -d)
    pushd deps/synapse-admin
    mv node_modules node_modules.bak
    cp -r $(readlink -f node_modules.bak) node_modules
    chmod +w node_modules
    yarn --offline run build
    popd

    runHook postBuild
  '';

  distPhase = ''
    runHook preDist

    mkdir -p $out
    cp -r deps/synapse-admin/build/* $out

    runHook postDist
  '';

  dontFixup = true;
  dontInstall = true;

  meta = with lib; {
    description = "Admin UI for Synapse Homeservers";
    homepage = "https://github.com/Awesome-Technologies/synapse-admin";
    license = licenses.asl20;
    platforms = platforms.all;
    maintainers = with maintainers; [ mkg20001 ma27 ];
  };
}
<commit_msg>synapse-admin: fix build after 7753a94a354<commit_after>{ lib
, stdenv
, fetchFromGitHub
, fetchzip
, mkYarnPackage
, baseUrl ? null
}:

mkYarnPackage rec {
  pname = "synapse-admin";
  version = "0.8.5";
  src = fetchFromGitHub {
    owner = "Awesome-Technologies";
    repo = pname;
    rev = version;
    sha256 = "sha256-0miHtEJ5e8MaqGc4ezPvwhGjoCZyOE7md0DUCC/ZOfk=";
  };

  yarnNix = ./yarn.nix;
  yarnLock = ./yarn.lock;
  packageJSON = ./package.json;

  NODE_ENV = "production";
  ${if baseUrl != null then "REACT_APP_SERVER" else null} = baseUrl;

  # error:0308010C:digital envelope routines::unsupported
  NODE_OPTIONS = "--openssl-legacy-provider";

  buildPhase = ''
    runHook preBuild

    export HOME=$(mktemp -d)
    pushd deps/synapse-admin
    mv node_modules node_modules.bak
    cp -r $(readlink -f node_modules.bak) node_modules
    chmod +w node_modules
    yarn --offline run build
    popd

    runHook postBuild
  '';

  distPhase = ''
    runHook preDist

    mkdir -p $out
    cp -r deps/synapse-admin/build/* $out

    runHook postDist
  '';

  dontFixup = true;
  dontInstall = true;

  meta = with lib; {
    description = "Admin UI for Synapse Homeservers";
    homepage = "https://github.com/Awesome-Technologies/synapse-admin";
    license = licenses.asl20;
    platforms = platforms.all;
    maintainers = with maintainers; [ mkg20001 ma27 ];
  };
}

{'fraction_non_alphanumeric': 0.11704545454545455, 'fraction_numerical': 0.024242424242424242, 'mean_word_length': 3.0259146341463414, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '662575', 'n_tokens_mistral': 1042, 'n_tokens_neox': 970, 'n_words': 215}